

Brachistochrone Curves


This notebook concerns the myphysicslab Brachistochrone simulation. This shows how the various curves used in that simulation were chosen. The goal is to find a variety of curves that start at the origin $(0, 0)$ and pass thru the point $(3, -2)$. The Brachistochrone simulation shows a ball sliding down each of the curves without friction, with gravity acting.

Author: Erik Neumann

a = 1;

x0 = 0; y0 = 0;

g1 = ParametricPlot[{{x0 + a (t - Sin[t]), -y0 - a (1 - Cos[t])}, {t, -t}}, {t, 0, Pi}, AspectRatio -> Automatic]



- Graphics -

Clear[a, t]

Solve[{3 == a (t - Sin[t]), 2 == a (1 - Cos[t])}, {a, t}]

Solve::tdep: The equations appear to involve the variables to be solved for in an essentially non-algebraic way.

Solve[{3 == a (t - Sin[t]), 2 == a (1 - Cos[t])}, {a, t}]


FindRoot[{3 == (2 / (1 - Cos[t])) (t - Sin[t])}, {t, Pi / 2}]

{t -> 3.06878}

a = 2 / (1 - Cos[3.06878])


1.00133

```
ParametricPlot[{{x0 + a (t - Sin[t]), -y0 - a (1 - Cos[t])}, {t, -2 t / 3},
  {t, -2 + (2 / 9) (t - 3) ^ 2}, {t, -(2 / 9) t ^ 2}}, {t, 0, 3.06878}, AspectRatio -> Automatic]
```




- Graphics -

```
ParametricPlot[{{x0 + a (t - Sin[t]), -y0 - a (1 - Cos[t])}, {t, -2 t / 3},
  {t, -2 + (2 / 9) (t - 3) ^ 2}, {t, -(2 / 9) t ^ 2}}, {t, 0, 8}, AspectRatio -> Automatic]
```



- Graphics -


```
Plot[{- (2 / 27) (x - 3) ^ 3 - 2}, {x, 0, 8}, AspectRatio -> Automatic, PlotRange -> All]
```



- Graphics -

```
k = 2.52699;
```

```
Plot[{- 2 - k + k Cosh[(x - 3) / k]}, {x, -1, 5}]
```




- Graphics -

```
FindRoot[{2 + q == q Cosh[3 / q]}, {q, 1}]
```

```
{q -> 2.52699}
```

```
ParametricPlot[
  {{x0 + a (t - Sin[t]), -y0 - a (1 - Cos[t])}, {t, -2 t / 3}, {t, -2 + (2 / 9) (t - 3) ^ 2},
   {t, - (2 / 9) t ^ 2}, {t, -2 - k + k Cosh[(t - 3) / k]}}, {t, 0, 8}, AspectRatio -> Automatic]
```



- Graphics -


The catenary with Cosh, is almost identical to the parabola... so not useful.

What about part of a circle?

```
Solve[{9 + (r - 2) ^ 2 == r ^ 2}, {r}]
```

```
{{r -> 13/4}}
```


```
ParametricPlot[
  {{x0 + a (t - Sin[t]), -y0 - a (1 - Cos[t])}, {t, -2 t / 3}, {t, -2 + (2 / 9) (t - 3) ^ 2},
   {t, - (2 / 9) t ^ 2}, {3 + (13 / 4) Cos[t], (13 / 4 - 2) + (13 / 4) Sin[t]}},
  {t, 0, 8}, AspectRatio -> Automatic, PlotRange -> {-3, 0.5},
  PlotStyle -> {Hue[0], Hue[0.1], Hue[0.2], Hue[0.3], Hue[0.4]}]
```



- Graphics -

The circle is really close to the brachistochrone... would be interesting to compare them.

```
ParametricPlot[{{x0 + a (t - Sin[t]), -y0 - a (1 - Cos[t])},
  {t, -2 t / 3}, {t, -2 + (2 / 81) (t - 3) ^ 4}}, {t, 0, 8}, AspectRatio -> Automatic,
  PlotRange -> {-3, 0.5}, PlotStyle -> {Hue[0], Hue[0.1], Hue[0.2], Hue[0.3], Hue[0.4]}]
```




- Graphics -

x^4 comes below the brachistochrone, but starts above it... that's probably good enough.
Part of a cardioid might be below the brachistochrone the entire way...

a

1.00133

```
ParametricPlot[{{x0 + a (t - Sin[t]), -y0 - a (1 - Cos[t])}, {t, -2 t / 3},
  {a (t - Sin[t]), 2 a (-1 + (1 - ((1 - Cos[t]) / 2)) ^ 2)}}, {t, 0, 8}, AspectRatio -> Automatic,
  PlotRange -> {-3, 1}, PlotStyle -> {Hue[0], Hue[0.1], Hue[0.2], Hue[0.3], Hue[0.4]}]
```



- Graphics -


The above shows taking the brachistochrone and "squaring it", to get a version that drops more quickly and below the brachistochrone the entire way.

I'm now looking for a version of the downward opening parabola. The problem with the one shown earlier is that the slope at the starting point (0,0) is zero, therefore the ball doesn't move at all! So I need to find a parabola that is close to that one, but with a non-zero slope at (0,0). I'm decided on a parabola of the form $f(x) = -k(x + 0.2)^2 + j$, it should pass through points (0,0) and (3, -2), so I can solve for k, j.

```
Solve[{-k (0 + 0.2) ^ 2 + j == 0, -k (3 + 0.2) ^ 2 + j == -2}, {k, j}]
```

```
{{k -> 0.196078, j -> 0.00784314}}
```

```
ParametricPlot[{{x0 + a (t - Sin[t]), -y0 - a (1 - Cos[t])},
  {t, -2 t / 3}, {t, -2 + (2 / 9) (t - 3) ^ 2}, {t, -0.196078 (t + 0.2) ^ 2 + 0.00784314},
  {3 + (13 / 4) Cos[t], (13 / 4 - 2) + (13 / 4) Sin[t]}}, {t, 0, 8}, AspectRatio -> Automatic,
  PlotRange -> {-3, 0.5}, PlotStyle -> {Hue[0], Hue[0.1], Hue[0.2], Hue[0.3], Hue[0.4]}]
```




- Graphics -

That parabola is still starting really slow... here is one that works a little better.

```
Solve[{-k (0 + 0.5) ^ 2 + j == 0, -k (3 + 0.5) ^ 2 + j == -2}, {k, j}]
```

```
{{k -> 0.166667, j -> 0.0416667}}
```

```
ParametricPlot[{{x0 + a (t - Sin[t]), -y0 - a (1 - Cos[t])},
  {t, -2 t / 3}, {t, -2 + (2 / 9) (t - 3) ^ 2}, {t, -0.166667 (t + 0.5) ^ 2 + 0.0416667},
  {3 + (13 / 4) Cos[t], (13 / 4 - 2) + (13 / 4) Sin[t]}}, {t, 0, 8}, AspectRatio -> Automatic,
  PlotRange -> {-3, 0.5}, PlotStyle -> {Hue[0], Hue[0.1], Hue[0.2], Hue[0.3], Hue[0.4]}]
```



- Graphics -

Next, I'm trying to solve the simultaneous equations to find the constant in the brachistochrone, according to the write up of the math I did...

```
FindRoot[{3 == -2 (t - Sin[t]) / (1 - Cos[t])}, {t, Pi}]
```

```
{t -> -3.06878}
```

```
2 / (1 - Cos[3.06878])
```

```
1.00133
```